# Plotting

---

**Exercise 1.1**

```
> restart;
```

**(a)**

```
> plot(2*exp(-t)*sin(30*t),t=-1..5);
```



This oscillates and dies away, like the motion of a guitar string after it has been plucked.

**(b)**

```
> plot(2*sin(20*t)+3*sin(21*t),t=0..30,numpoints=1000);
```



This oscillates rapidly (at the same frequency as $\sin(20.5\,t)$), with the size of the oscillations varying more slowly (like $\cos(t)$).

**(c)**

```
> plot(sin(x)+sin(3*x)/3+sin(5*x)/5+sin(7*x)/7,x=0..20);
```




Half the time this wiggles around near $y = 0.8$, and half the time it wiggles around near $y = -0.8$. It jumps quite rapidly between the two levels, and the jumps occur at multiples of Pi ( Pi = 3.14, 2 Pi = 6.28, 3 Pi = 9.42 and so on).

**(d)**

```
> plot(cos(Pi*x^2),x=-5..5);
```




The function oscillates between $-1$ and $1$, with the oscillations becoming more and more rapid as the absolute value of $x$ increases.

---

## Exercise 1.2

```
> restart;
> f := (x) -> (x^3-x)*(x^2-4/9)*(x^2-1/9);
```

$$f := x \mapsto (x^3 - x)\left(x^2 - \frac{4}{9}\right)\left(x^2 - \frac{1}{9}\right) \quad (1)$$

The graph is quite flat and close to zero between $x = -1$ and $x = 1$, crossing the $x$-axis seven times, at $x = -1, -\frac{2}{3}, -\frac{1}{3}, 0, \frac{1}{3}, \frac{2}{3}$ and $1$. The function increases rapidly towards $\infty$ for $1 < x$, and drops rapidly towards $-\infty$ for $x < -1$.

```
> plot(f(x),x=-2..2,-0.5..0.5);
```




The rate of growth is similar to $x^7$, and in fact $f(x)$ is indistinguishable from $x^7$ when $x$ is reasonably large.

```
> plot([f(x),x^7],x=-3..3);
```




```
> plot([f(x),x^7],x=-10..10);
```




To see the reason for this, just expand out $f(x)$. The leading term is $x^7$, and the remaining terms are multiples of $x^5$ or lower, so they are much smaller than $x^7$ when $x$ is large.

```
> expand(f(x));
```

$$x^7 - \frac{14}{9}\,x^5 + \frac{49}{81}\,x^3 - \frac{4}{81}\,x \qquad (2)$$

---

## Exercise 1.3

```
> restart;
> g := (x) -> (2*x-3)/(3*x-4);
```

$$g := x \mapsto \frac{2\,x-3}{3\,x-4} \qquad (3)$$

**(a)** Most of the time the graph is very flat, and close to $y = 0.7$. Somewhere close to $x = 1.3$, the graph climbs steeply to $\infty$, jumps down discontinuously to $-\infty$, and then climbs back up to around 0.7 again.

```
> plot(g(x),x=-10..10,-10..10);
```




```
> plot(g(x),x=-20..20,0..2,axes=boxed);
```




```
> plot(g(x),x=1..2,-2..4);
```




**(b)** The function is discontinuous at the point where the formula $g(x) = \frac{2x-3}{3x-4}$ involves division by zero, which means $3x-4=0$ or in other words $x = \frac{4}{3}$. This can be found graphically as the point where the vertical line crosses the axis in the picture above. Alternatively:

```
> discont(g(x),x);
```

$$\left\{\frac{4}{3}\right\} \quad (4)$$

Here we replot the graph, skipping over the discontinuity:

```
>   plot(g(x),x=-3..5,-10..10,discont=true);
```




**(c)**

```
> plot([g(x),5],x=0..3,-10..10,discont=true);
```




This picture shows that the graph does not cross the line $y = \frac{2}{3}$

```
> plot([g(x),2/3],x=-20..20,0..1,discont=true);
```




To see this algebraically, we first find the general formula for the point where the curve crosses the line $y = a$:

```
> solve(g(x)=a,{x});
```

$$\left\{x = \frac{4\,a - 3}{-2 + 3\,a}\right\} \qquad \textbf{(5)}$$

The formula involves division by zero when $a = \frac{2}{3}$, indicating that the curve does not in fact cross the

line

$y = \frac{2}{3}$.

**(d)**

```
> limit(g(x),x=infinity);
```

$$\frac{2}{3} \tag{6}$$

```
> limit(g(x),x=-infinity);
```

$$\frac{2}{3} \tag{7}$$

This is of course the same answer as in (c). As $x$ approaches $\infty$, the function $g(x)$ approaches $\frac{2}{3}$ from below, but never reaches it. As $x$ approaches $-\infty$, the function approaches $\frac{2}{3}$ from above, but again never reaches it.

---

## Exercise 1.4

```
> restart;
> f := (x)->(1-exp(x)+exp(2*x)-exp(3*x))/sin(x);
```

$$f := x \mapsto \frac{1 - e^{x} + e^{2x} - e^{3x}}{\sin(x)} \tag{8}$$

```
> f(0);
Error, (in f) numeric exception: division by zero
> plot(f(x),x=-1..1);
```




Although the formula for $f(x)$ does not make sense when $x = 0$, the only reasonable definition is $f(0) = -2$, as we see from the graph. Here is a non-graphical way to get Maple to produce this answer:

```
> limit(f(x),x=0);
```

$$-2 \tag{9}$$

## Exercise 1.5

```
> restart;
```

The Bessel function is even (ie $J_2(-x) = J_2(x)$). It oscillates quite regularly, with amplitude decreasing slowly as $x$ approaches $\infty$ or $-\infty$.

```
> plot(BesselJ(2,x),x=-50..50);
```




For small $x$, the function $J_2(x)$ is very close to $\frac{x^2}{8}$.

```
> plot([BesselJ(2,x),x^2/5],x=-0.1..0.1);
```




```
> plot([BesselJ(2,x),x^2/8],x=-0.1..0.1);
```




This plot has glitches because Maple has not calculated enough points.

```
> plot(BesselJ(2,x),x=10..300);
```




We can cure this with the **numpoints** option.

```
> plot(BesselJ(2,x),x=10..300,numpoints=200);
```




```
> pic := %:
```

Here are several attempts to find the 'envelope' of the graph. The final attempt is $y = 0.8\,x^{-\frac{1}{2}}$, which fits very nicely with the peaks of the waves.

```
> with(plots):
```

```
> display(pic,plot(6*x^(-1),x=10..300,colour=blue));
```




```
> display(pic,plot(0.3*x^(-1/3),x=10..300,colour=blue));
```




```
> display(pic,plot(0.5*x^(-1/2),x=10..300,colour=blue));
```




```
> display(pic,plot(0.8*x^(-1/2),x=10..300,colour=blue));
```




This plot shows that $J_2(x)$ has (to a very good approximation) the same frequency as $\sin(x)$.

```
> plot([BesselJ(2,x),sin(x+Pi/4)],x=10..100);
```




By combining the above results, we see that the following function should be a good approximation to $J_2(x)$ for large $x$. The plot below shows this graphically.

```
> f := (x) -> -4*sin(x+Pi/4)/(5*sqrt(x));
```

$$f := x \mapsto -\frac{4\sin\left(x+\frac{\pi}{4}\right)}{5\sqrt{x}} \qquad \textbf{(10)}$$

```
> plot([BesselJ(2,x),f(x)],x=1..100);
```




---

**Exercise 1.6**

```
> g := (a,x) -> (x-1-sin(a/4))*(x-1-cos(a/4))*
                (x+1-sin(a/4))*(x+1-cos(a/4))/(1+x^4);
```

$$g := (a, x) \mapsto \frac{\left(x - 1 - \sin\left(\frac{a}{4}\right)\right)\left(x - 1 - \cos\left(\frac{a}{4}\right)\right)\left(x + 1 - \sin\left(\frac{a}{4}\right)\right)\left(x + 1 - \cos\left(\frac{a}{4}\right)\right)}{1 + x^4} \quad \textbf{(11)}$$

**(a)** The graph $y = g(2, x)$ approaches $y = 1$ very closely when $x$ is very large (positive or negative). Close to $x = 0$ it climbs quickly to about $y = 2.5$, then drops very quickly to about $y = 0$, then climbs again to just below $y = 1$. The graphs for $g(3, x)$ and $g(4, x)$ look very similar on this scale.

```
> plot(g(2,x),x=-500..500);
```



```
> plot(g(3,x),x=-500..500);
```




```
> plot([g(4,x),1],x=-500..500);
```




**(b)**

```
> plot(g(2,x),x=-1..2);
```




```
> plot(g(4,x),x=-1..2);
```




```
> plot(g(3,x),x=-1..2);
```




```
> plot(g(3,x),x=-0.4..-0.2,-0.01..0,numpoints=1000);
```




It works out that the graph of $g(\mathrm{Pi}, x) = g(3.1415927, x)$ stays above the $x$-axis, just touching it in two places. We first show this graphically:

```
> plot(g(Pi,x),x=-1..3);
```




We now zoom in to the first point where the curve touches the axis:

```
> plot(g(Pi,x),x=-.292899..-.292885,-1e-10..1e-10,numpoints=1000);
```




To see why this works algebraically, note that $\sin\left(\frac{\mathrm{Pi}}{4}\right) = \frac{\sqrt{2}}{2}$ and $\cos\left(\frac{\mathrm{Pi}}{4}\right) = \frac{\sqrt{2}}{2}$. Putting this in the definition of $g(a, x)$ gives the following:

```
> g(Pi,x);
```

$$\frac{\left(x - 1 - \frac{\sqrt{2}}{2}\right)^2 \left(x + 1 - \frac{\sqrt{2}}{2}\right)^2}{x^4 + 1} \quad \mathbf{(12)}$$

As the square or fourth power of any real number is nonnegative, this formula shows that $0 \le g(\mathrm{Pi}, x)$ for all $x$.

**(c)** It turns out that $1 - 8\, g(a, 0) = \cos(a)$. To see this graphically, compare the two plots below.

```
> plot(1-8*g(a,0),a=-10..10);
```




```
> plot(cos(a),a=-10..10);
```




However, Maple's simplification commands are not clever enough to work this out:

```
> 1-8*g(a,0);
```

$$1-8\left(-1-\sin\left(\frac{a}{4}\right)\right)\left(-1-\cos\left(\frac{a}{4}\right)\right)\left(1-\sin\left(\frac{a}{4}\right)\right)\left(1-\cos\left(\frac{a}{4}\right)\right) \quad \textbf{(13)}$$

```
> simplify(1-8*g(a,0));
```

$$8\cos\left(\frac{a}{4}\right)^4-8\cos\left(\frac{a}{4}\right)^2+1 \quad \textbf{(14)}$$

If we give Maple a hint by asking it to simplify $1-8\,g(a,0)-\cos(a)$ instead, it does succeed in working out that this is zero:

```
> simplify(1-8*g(a,0)-cos(a));
```

$$0 \quad \textbf{(15)}$$

Here is a trick that persuades Maple to convert $1-8\,g(a,0)$ to $\cos(a)$. It is based on De Moivre's Theorem, which will be covered in the course on complex numbers.

```
> simplify(convert(1-8*g(a,0),exp));
```

$$\cos(a) \quad \textbf{(16)}$$

To do this simplification by hand, note that $\left(-1-\cos\left(\frac{a}{4}\right)\right)\left(1-\cos\left(\frac{a}{4}\right)\right)=\cos\left(\frac{a}{4}\right)^2-1$, which is the same as $-\sin\left(\frac{a}{4}\right)^2$. Similarly, we have $\left(-1-\sin\left(\frac{a}{4}\right)\right)\left(1-\sin\left(\frac{a}{4}\right)\right)=-\cos\left(\frac{a}{4}\right)^2$. Putting these together, we see that $8\,g(a,0)=8\sin\left(\frac{a}{4}\right)^2\cos\left(\frac{a}{4}\right)^2$, which is the same as $2\left(2\sin\left(\frac{a}{4}\right)\cos\left(\frac{a}{4}\right)\right)^2$. Using the standard formula $\sin(2\,\theta)=2\sin(\theta)\cos(\theta)$, this simplifies to $2\sin\left(\frac{a}{2}\right)^2$. We can then use the standard formula $\cos(2\,\phi)=1-2\sin(\phi)^2$ to deduce that $1-8\,g(a,0)=\cos(a)$.

**(d)**

```
> plot(g(a,0.5),a=-40..40);
```




```
> slope := diff(g(a,0.5),a);
```

$$slope := -0.2352941176\cos\left(\frac{a}{4}\right)\left(-0.5-\cos\left(\frac{a}{4}\right)\right)\left(1.5-\sin\left(\frac{a}{4}\right)\right)\left(1.5-\cos\left(\frac{a}{4}\right)\right) + 0.2352941176\left(-0.5-\sin\left(\frac{a}{4}\right)\right)\sin\left(\frac{a}{4}\right)\left(1.5-\sin\left(\frac{a}{4}\right)\right)\left(1.5-\cos\left(\frac{a}{4}\right)\right) - 0.2352941176\left(-0.5-\sin\left(\frac{a}{4}\right)\right)\left(-0.5-\cos\left(\frac{a}{4}\right)\right)\cos\left(\frac{a}{4}\right)\left(1.5-\cos\left(\frac{a}{4}\right)\right) + 0.2352941176\left(-0.5-\sin\left(\frac{a}{4}\right)\right)\left(-0.5-\cos\left(\frac{a}{4}\right)\right)\left(1.5-\sin\left(\frac{a}{4}\right)\right)\sin\left(\frac{a}{4}\right) \quad \textbf{(17)}$$

One of the tall peaks is at $x = \mathrm{Pi}$.

```
> fsolve(slope=0,a=3);
```

$$3.141592654 \quad \textbf{(18)}$$

The others are at $x = -7\ \mathrm{Pi}$ and $x = 9\ \mathrm{Pi}$.

```
> fsolve(slope=0,a=-22);
```

$$-21.99114858 \quad \textbf{(19)}$$

```
> evalf(%/Pi);
```

$$-7.000000001 \quad \textbf{(20)}$$

```
> fsolve(slope=0,a=29);
```

$$28.27433388 \quad \textbf{(21)}$$

```
> evalf(%/Pi);
```

$$8.999999998 \quad \textbf{(22)}$$

**(e)**

```
> plot3d(g(a,x),a=-20..20,x=-15..15,
         axes=boxed,grid=[100,100]);
```



2
1
0
−1
−15
−10
−5
0
5
10
20
x
10
0
−10
−20
a